
Materials in these notes are mainly from Kaldewaij. Some
examples are adapted from the course CSci 550: Program
Semantics and Derivation taught by Prof. H. Conrad
Cunningham, University of Mississippi.

SOME NOTES ON DEFINEDNESS



ASSIGNMENT REVISITED

• Recall the weakest precondition for assignments:

wp (x := E) P = P[x\E] .

• That is not the whole story... since we have to be sure that
E is defined!

DEFINEDNESS

• In our current language, given expression E there is a
systematic (inductive) definition on what needs to be
proved to ensure that E is defined. Let’s denote it by def E.

• We will not go into the detail but give examples.
• For example, if there is division in E, the denominator
must not be zero.

• def (x+ y / (z+ x)) = (z+ x ̸= 0).
• def (x+ y / 2) = (2 ̸= 0) = True.

WEAKEST PRECONDITION

• A more complete rule:

wp (x := E) P = P[x\E] ∧ def E .

• In fact, all expressions need to be defined. E.g.

wp (if B0 → S0 | B1 → S1 fi) P =

B0 ⇒ wp S0 P ∧ B1 ⇒ wp S1 P ∧ (B0 ∨ B1) ∧
def B0 ∧ def B1 .

HOW COME WE HAVE NEVER MENTIONED SO?

• How come we have never mentioned so?
• The first partial operation we have used was division. And
the denominator was usually a constant (namely, 2!).

ARRAY BOUND

• Array indexing is a partial operation too — we need to be
sure that the index is within the domain of the array.

• Let A : array [M..N) of Int and let I be an expression. We
define def (A[I]) = def I ∧ M ⩽ I< N.

• E.g. given A : array [0..N) of Int,
• def (A[x / z] + A[y]) = z ̸= 0 ∧ 0 ⩽ x / z< N ∧ 0 ⩽ y< N.
• wp (s := s ↑ A[n]) P = P[s\s ↑ A[n]] ∧ 0 ⩽ n< N.

• We never made it explicit, because conditions such as
0 ⩽ n< N were usually already in the invariant/guard and
thus discharged immediately.

ARRAY ASSIGNMENT



ARRAY ASSIGNMENT

• So far, all our arrays have been constants — we read from
the arrays but never wrote to them!

• Consider a : array [0..2) of Int, where a[0] = 1 and a[1] = 1.
• It should be true that

{a[0] = 1 ∧ a[1] = 1}
a[a[1]] := 0
{a[a[1]] = 1} .

• However, if we use the previous wp,

wp (a[a[1]] := 0) (a[a[1]] = 1)
≡ (a[a[1]] = 1)[a[a[1]]\0]
≡ 0 = 1
≡ False .

• What went wrong?

ANOTHER COUNTEREXAMPLE

• For a more obvious example where our previous wp does
not work for array assignment:

• wp (a[i] := 0) (a[2] ̸= 0) appears to be a[2] ̸= 0, since a[i]
does not appear (verbatim) in a[2] ̸= 0.

• But what if i = 2?

ARRAYS AS FUNCTIONS

• An array is a function. E.g. a : array [0..N) of Bool is a
function Int→ Bool whose domain is [0..N).

• Indexing a[n] is function application.
• Some textbooks use the same notation for function
application and array indexing.

• (Could that have been a better choice for this course?)

FUNCTION ALTERATION

• Given f : A→ B, let (f :x )e) denote the function that maps
x to e, and otherwise the same as f.

(f :x )e) y = e , if x = y;
= f y , otherwise.

• For example, given f x = x2, (f :1 )−1) is a function such
that

(f :1 )−1) 1 = −1 ,

(f :1 )−1) x = x2 , if x ̸= 1.

wp FOR ARRAY ASSIGNMENT

• Key: assignment to array should be understood as altering
the entire function.

• Given a : array [M..N) of A (for any type A), the updated
rule:

wp (a[I] := E) P = P[a\(a : I )E)] ∧
def (a[I]) ∧ def E .

• In our examples, def (a[I]) and def E can often be
discharged immediately. For example, the boundary check
M ⩽ I< N can often be discharged soon. But do not forget
about them.

THE EXAMPLE

• Recall our example

{a[0] = 1 ∧ a[1] = 1}
a[a[1]] := 0
{a[a[1]] = 1} .

• We aim to prove

a[0] = 1 ∧ a[1] = 1 ⇒
wp (a[a[1]] := 0) (a[a[1]] = 1) .

Assume a[0] = 1 ∧ a[1] = 1.

wp (a[a[1]] := 0) (a[a[1]] = 1)
≡ { def. of wp for array assignment }
(a :a[1] )0)[(a :a[1] )0)[1]] = 1

≡ { assumption: a[1] = 1 }
(a :1 )0)[(a :1 )0)[1]] = 1

≡ { def. of alteration: (a :1 )0)[0] = 0 }
(a :1 )0)[0] = 1

≡ { def. of alteration: (a :1 )0)[0] = a[0] }
a[0] = 1

≡ { assumption: a[0] = 1 }
True .

RESTRICTIONS

• In this course, parallel assignments to arrays are not
allowed.

• This is done to avoid having to define what the following
program ought to do:

x, y := 0, 0;
a[x],a[y] := 0, 1

• It is possible to give such programs a definition (e.g.
choose an order), but we prefer to keep it simple.

TYPICAL ARRAY MANIPULATION IN A
LOOP



EXAMPLE: ALL ZEROS

Consider:

con N : Int {0 ⩽ N}
var h : array [0..N) of Int
allzeros
{⟨∀i : 0 ⩽ i< N : h[i] = 0⟩}

THE USUAL DRILL

con N : Int {0 ⩽ N}
var h : array [0..N) of Int
var n : Int

n := 0
{⟨∀i : 0 ⩽ i< n : h[i] = 0⟩ ∧ 0 ⩽ n ⩽ N,
bnd : N− n}

do n ̸= N→ ?

n := n+ 1
od
{⟨∀i : 0 ⩽ i< N : h[i] = 0⟩}

• The calculation can certainly be generalised.
• Given a function H : Int→ A, and suppose we want to
establish

⟨∀i : 0 ⩽ i< N : h[i] = H i⟩ ,

where H does not depend on h (e.g, h does not occur free
in H).

• Let P n = 0 ⩽ n< N ∧ ⟨∀i : 0 ⩽ i< n : h[i] = H i⟩).
• We aim to establish P (n+ 1), given P n ∧ n ̸= N.

• One can prove the following:

{P n ∧ n ̸= N ∧ E = H n}
h[n] := E
{P (n+ 1)} ,

• which can be used in a program fragment...

{P 0}
n := 0
{P n,bnd : N− n}
do n ̸= N→

{ establish E = H n }
h[n] := E
n := n+ 1

od
{⟨∀i : 0 ⩽ i< N : h[i] = H i⟩}

• Why do we need E? Isn’t E simply H n?
• In some cases H n can be computed in one expression. In
such cases we can simply do h[n] := H n.

• In some cases E may refer to previously computed results
— other variables, or even h.

• Yes, E may refer to h while H does not. There are such
examples in the Practicals.

EXAMPLE: HISTOGRAM

Consider:

con N : Int {0 ⩽ N}; X : array [0..N) of Int
{⟨∀i : 0 ⩽ i< N : 1 ⩽ X[i] ⩽ 6⟩}
var h : array [1..6] of Int
histogram
{⟨∀i : 1 ⩽ i ⩽ 6 : h[i] =
⟨#k : 0 ⩽ k< N : X[k] = i⟩⟩}

THE PROGRAM

Let P n ≡ ⟨∀i : 1 ⩽ i ⩽ 6 : h[i] = ⟨#k : 0 ⩽ k< n : X[k] = i⟩⟩.

con N : Int {0 ⩽ N}; X : array [0..N) of Int
{⟨∀i : 0 ⩽ i< N : 1 ⩽ X[i] ⩽ 6⟩}
var h : array [1..6] of Int
var n : Int

n := 1
do n ̸= 7 → h[n] := 0;n := n+ 1 od
{P 0}

n := 0
{P n ∧ 0 ⩽ n ⩽ N,bnd : N− n}
do n ̸= N→ h[X[n]] := h[X[n]] + 1

n := n+ 1
od
{⟨∀i : 1 ⩽ i ⩽ 6 : h[i] =
⟨#k : 0 ⩽ k< N : X[k] = i⟩⟩}

• swap h E F does not always literally “swaps the values.”
For example, it is not always the case that

{h[E] = X} swap h E F {h[F] = X} .

• Consider h[0] = 0 ∧ h[1] = 1. This does not hold:

{h[h[0]] = 0} swap h (h[0]) (h[1]) {h[h[1]] = 0} .

• In fact, after swapping we have h[0] = 1 ∧ h[1] = 0, and
hence h[h[1]] = 1.

A SIMPLER CASE

• However, when h does not occur free in E and F, we do
have

({⟨∀i : i ̸= E ∧ i ̸= F : h[i] = H i⟩} ∧
h[E] = X ∧ h[F] = Y)

swap h E F
({⟨∀i : i ̸= E ∧ i ̸= F : h[i] = H i⟩} ∧
h[E] = Y ∧ h[F] = X) .

• It is a convenient rule we use when reasoning about
swapping.

• Note that, in the rule above, E and F are expressions, while
X, Y, H are logical variables.

NOTE: KALDEWAIJ’S SWAP

• Kaldewaij defined swap h E F as an abbreviation of

|[ var r; r := h[E];h[E] := h[F];h[F] := r ]| ,

• where r is a fresh name and |[...]| denotes a program block
with local constants and variables. We have not used this
feature so far.

• I do not think this definition is correct, however. The
definition would not behave as we expect if F refers to h[E].

THE DUTCH NATIONAL FLAG

• Let RWB = {R,W,B} (standing respecively for red, white,
and blue).

con N : Int {0 ⩽ N}
var h : array [0..N) of RWB
var r,w : Int
dutch_national_flag
{0 ⩽ r ⩽ w ⩽ N ∧
⟨∀i : 0 ⩽ i< r : h[i] = R⟩ ∧
⟨∀i : r ⩽ i< w : h[i] = W⟩ ∧
⟨∀i : w ⩽ i< N : h[i] = B⟩ ∧}

• The program shall manipulate h only by swapping.
• Denote the postcondition by Q.

WHITE

• The case for white is the easiest, since

P0 ∧ P1 ∧ h[w] = W⇒
(P0 ∧ P1)[w\w+ 1] .

• It is sufficient to let Sw be simply w := w+ 1.

BLUE

• We have

{Pr ∧ Pw ∧ Pb ∧ w< b ∧ h[w] = B}
swap h w (b− 1)
{Pr ∧ Pw ∧ Pb ∧ w< b ∧ h[b− 1] = B}
b := b− 1
{Pr ∧ Pw ∧ Pb ∧ w ⩽ b}

• Thus we choose swap h w (b− 1);b := b− 1 as Sb.

RED

• Precondition: Pr ∧ Pw ∧ Pb ∧ w< b ∧ h[w] = R.
• It appears that swap h w r establishes P[w\w+ 1]. But we
have to see what h[r] is before we can increment r.

• Pw implies r< w⇒ h[r] = W. Equivalently, we have
r = w ∨ h[r] = W.

RED: CASE r = w

• We have

{Pr ∧ Pw ∧ Pb ∧ r = w< b ∧ h[w] = R}
swap h w r
{Pr ∧ Pw ∧ Pb ∧ w< b ∧ h[r] = R}
r,w := r+ 1,w+ 1
{Pr ∧ Pw ∧ Pb ∧ r = w ⩽ b}

RED: CASE h[r] = W

• We have

{Pr ∧ Pw ∧ Pb ∧ w< b ∧ h[r] = W ∧ h[w] = R}
swap h w r
{Pr ∧ h[r] = R ∧ ⟨∀i : r+ 1 ⩽ i< w : h[i] = W⟩ ∧
h[w] = W ∧ Pb ∧ w< b}

r,w := r+ 1,w+ 1
{Pr ∧ Pw ∧ Pb ∧ r = w ⩽ b}

• In both cases, swap h w r; r,w := r+ 1,w+ 1 is a valid
choice.

•

con K,N : Int {0 ⩽ K< N}
var h : array [0..N) of A
{⟨∀i : 0 ⩽ i< N : h[i] = H[i]⟩}
rotation
{⟨∀i : 0 ⩽ i< N : h[(i+ K)mod N] = H[i]⟩} .

• To eliminate mod, the postcondition can be rewritten as:

⟨∀i : 0 ⩽ i< N− K : h[i+ K] = H[i]⟩ ∧
⟨∀i : N− K ⩽ i< N : h[i+ K− N] = H[i]⟩ .

• Or, h[K..N) = H[0..N− K) ∧ h[0..K) = H[N− K..N).

ABSTRACT NOTATIONS

• For this problem we benefit from using more abstract
notations.

• Segments of arrays can be denoted by variables. E.g.
X = H[0..N− K) and Y = H[N− K..N).

• Concatenation of arrays are denoted by juxtaposition. E.g.
H[0..N) = XY.

• Empty sequence is denoted by [ ].
• Length of a sequence X is denoted by l X.

• Specification:

{h = XY}
rotation
{h = YX}

• When l X = l Y we can establish the postcondition easily —
just swap the corresponding elements.

• Denote swapping of equal-lengthed array segments by
SWAP X Y.

THINKING LENGTHS

• When l X< l Y, h can be written as h = XUV,
• where l U = l X and UV = Y.
• Task:

{h = XUV ∧ l U = l X}
rotation
{h = UVX}

• Strategy:

{h = XUV ∧ l U = l X}
SWAP X U
{h = UXV}
??

{h = UVX}

• The part ?? shall transform XV into VX — a problem having
the same form as the original!

• Some (including myself) would then go for a recursive
program. But there is another possibility.

LEADING TO AN INVARIANT...

• Consider the symmetric case where l X> l Y.

{h = UVY ∧ l V = l Y}
SWAP V Y
{h = UYV}
??

{h = YUV}

• In general, the array is of them form AUVB, where UV
needs to be transformed into VU, while A and B are parts
that are done.

THE INVARIANT

• Strategy:

{h = XY}
A,U, V,B := [ ], X, Y, [ ]
{h = AUVB ∧ YX = AVUB,bnd : l U+ l V}
do U ̸= [ ] ∧ V ̸= [ ] → ...od
{h = YX}

• Call the invariant P. Intuitively it means “currently the
array is AUVB, and if we exchange U and V, we are done.”

• Note the choice of guard: P ∧ (U = [ ] ∧ V = [ ]) ⇒ h = YX.

AN ABSTRACT PROGRAM

A,U, V,B := [ ], X, Y, [ ]
{h = AUVB ∧ YX = AVUB,bnd : l U+ l V}
do U ̸= [ ] ∧ V ̸= [ ] →
if l U ⩾ l V→ -- l U1 = l V
{h = AU0U1VB ∧ YX = AVU0U1B}
SWAP U1 V
{h = AU0VU1B ∧ YX = AVU0U1B}
U,B := U0,U1B
{h = AUVB ∧ YX = AVUB}

| l U ⩽ l V→ -- l V0 = l U
{h = AUV0V1B ∧ YX = AV0V1UB}
SWAP U V0
{h = AV0UV1B ∧ YX = AV0V1UB}
A, V := AV0, V1
{h = AUVB ∧ YX = AVUB}

fi
od

REPRESENTING THE SEQUENCES

• Introduce a,b, k, l : Int.
• A = h[0..a);
• U = h[a..a+ k), hence l U = k;
• V = h[b− l..b), hence l V = l;
• B = h[b..N).
• Additional invariant: a+ k = b− l.
• Why having both k and l? We will see later.

A CONCRETE PROGRAM

• Represented using indices:
a, k, l,b := 0,N− K, K,N
do k ̸= 0 ∧ l ̸= 0 →
if k ⩾ l→ SWAP (b− l) l (−l)

k,b := k− l,b− l
| k ⩽ l→ SWAP a k k

a, l := a+ k, l− k
fi

od
• where SWAP x num off abbreviates

|[ var n : Int
n := x
do n ̸= x+ num→ swap h n (n+ off)

n := n+ 1
od

]|

• that is, starting from index x, swap num elements with
those off positions away.

GREATEST COMMON DIVISOR

• To find out the number of swaps performed, we use a
variable t to record the number of swaps.

• If we keep only computation related to t, k, and l:

k, l, t := N− K, K, 0
do k ̸= 0 ∧ l ̸= 0 →
if k ⩾ l→ t := t+ l; k := k− l
| k ⩽ l→ t := t+ k; l := l− k
fi

od

• Observe: the part concerning k and l resembles
computation of greatest common divisor.

• In fact, gcd k l = gcd N (N− K), which is gcd N K.
• When the program terminates, k+ l = gcd N K.
• It’s always true that t+ k+ l = N.
• Therefore, the total number of swaps is
t = N− (k+ l) = N− gcd N K.

